$\mathbb{A}^1$-BROUWER DEGREES IN MACAULAY2

NIKITA BORISOV, THOMAS BRAZELTON, FRENLY ESPINO, THOMAS HAGEDORN, ZHAOBO HAN, JORDY LOPEZ GARCIA, JOEL LOUWSMA, WERN JUIN GABRIEL ONG, AND ANDREW R. TAWFEEK

Abstract. We describe the *Macaulay2* package `A1BrouwerDegrees` for computing local and global $\mathbb{A}^1$-Brouwer degrees and studying symmetric bilinear forms over the complex numbers, the real numbers, the rational numbers, and finite fields of characteristic not equal to 2.

1. Introduction

In $\mathbb{A}^1$-homotopy theory, the $\mathbb{A}^1$-*Brouwer degree* provides an algebro-geometric analogue of the classical Brouwer degree from differential topology. Morel's $\mathbb{A}^1$-degree homomorphism identifies the zeroth stable stem of the motivic sphere spectrum with the *Grothendieck–Witt ring* of symmetric bilinear forms over a field [18, Corollary 1.24]. Given an endomorphism of affine space with an isolated rational zero, work of Kass and Wickelgren [10] identifies its local $\mathbb{A}^1$-Brouwer degree with the Eisenbud–Khimshiashvili–Levine signature form [5, 7], which was used to compute local Brouwer degrees in real differential topology. Work of Bachmann and Wickelgren [2] extends this work, identifying the $\mathbb{A}^1$-Brouwer degree with a quadratic Grothendieck–Serre duality form.

In $\mathbb{A}^1$-*enumerative geometry* [11, 15] (see [3, 20] for an overview), the $\mathbb{A}^1$-Brouwer degree has found a wealth of applications, recently including [1, 9, 8]. For instance, via McKean's Bézout theorem, the $\mathbb{A}^1$-Brouwer degree can be understood as a quadratically enriched analogue of intersection multiplicity, often encoding deeper geometric information than was available over the algebraically closed fields – with other invariants of the quadratic form over k capturing field-specific arithmetic data [16].

Recent work of the second-named author, McKean, and Pauli [4] provides tractable formulas for computing $\mathbb{A}^1$-Brouwer degrees as *Bézoutian bilinear forms*. In the `A1BrouwerDegrees` package, we implement these methods in *Macaulay2* [6] over the fields $\mathbb{C}$, $\mathbb{R}$, $\mathbb{Q}$, and $\mathbb{F}_q$ (for q odd) and provide a suite of tools whose capabilities include:

(1) computing $\mathbb{A}^1$-Brouwer degrees (both local and global) for endomorphisms of affine space;[1]
(2) decomposing symmetric bilinear forms into their isotropic and anisotropic parts; and
(3) extracting invariants of symmetric bilinear forms (rank, signature, discriminant, Hasse–Witt invariants).

Remark 1. The current scope of this package is the complex numbers, the real numbers, the rational numbers, and finite fields of characteristic not equal to 2, so all of our algorithms are

2020 *Mathematics Subject Classification.* 14F42, 14-04, 68W30, 11E04, 55M25, 14N10.

[1] Due to $\mathbb{R}$ being an inexact field, $\mathbb{A}^1$-Brouwer degrees over $\mathbb{R}$ have to be computed over $\mathbb{Q}$ and base-changed to $\mathbb{R}$.

implicitly taking place in these settings. We hope to expand the scope of these algorithms in future work.

In Section 2, we provide a rapid introduction to the theory of symmetric bilinear forms, highlighting the capacity of our package to build forms, check isomorphisms, and decompose forms. In Section 3, we discuss local and global $\mathbb{A}^1$-Brouwer degrees and provide some computational examples, including quadratically enriched intersection multiplicity of real curves, the $\mathbb{A}^1$-Euler characteristic of the Grassmannian $\mathrm{Gr}(2,4)$ (following [4, Section 8.2]), and local computations for 27 lines on a cubic surface (following [11, 19]).

1.1. **Software availability.** The software documented here is available in versions 1.23 and later of *Macaulay2* as the `A1BrouwerDegrees` package.

2. The Grothendieck–Witt ring

For this entire section, we assume k is a field of characteristic not equal to 2. We say a bilinear form $\beta\colon V \times V \to k$ is *symmetric* if $\beta(v,w) = \beta(w,v)$ for all $v, w \in V$. We say β is *non-degenerate* if $\beta(v,-)\colon V \to k$ is identically zero if and only if $v = 0$.

Definition 2. Let $\beta\colon V \times V \to k$ be a symmetric bilinear form, and choose a basis $e_1, \ldots, e_n$ for V. We define the *Gram matrix* of β in the basis $\{e_i\}_{i=1}^n$ to be the symmetric matrix with entries $\beta(e_i, e_j)$.

Remark 3. Non-degeneracy of β is equivalent to the statement that the determinant of a Gram matrix in any basis is nonzero. A change of basis for V corresponds to the associated Gram matrices being congruent.

Given two symmetric bilinear forms $\beta_i\colon V_i \times V_i \to k$ for $i = 1, 2$, we can define their sum and product:

$$\begin{aligned}(\beta_1 \oplus \beta_2)&\colon (V_1 \oplus V_2) \times (V_1 \oplus V_2) \to k \\ (\beta_1 \otimes \beta_2)&\colon (V_1 \otimes V_2) \times (V_1 \otimes V_2) \to k.\end{aligned} \tag{1}$$

On Gram matrices, these operations are given by direct sum and tensor product, respectively.

Definition 4. The *Grothendieck–Witt ring* $\mathrm{GW}(k)$ is the group completion of the semiring of isomorphism classes of non-degenerate symmetric bilinear forms over k.

Example 5. Any non-degenerate symmetric bilinear form over an algebraically closed field admits a basis in which its Gram matrix is the identity; therefore rank determines an isomorphism $\mathrm{GW}(\mathbb{C}) \cong \mathbb{Z}$. For further computations of Grothendieck–Witt rings, we refer the reader to [14, Chapter II].

When the field k is the complex numbers, the real numbers, the rational numbers, or a finite field (of characteristic not 2), we define a type called `GrothendieckWittClass` that encodes the class $[\beta] \in \mathrm{GW}(k)$ of a symmetric bilinear form β. Grothendieck–Witt classes can be constructed from Gram matrices via the `makeGWClass` method.

```
i1 : needsPackage "A1BrouwerDegrees";
i2 : M = matrix(QQ, {{1,3},{3,7}});
o2 : Matrix ℚ^2 <--- ℚ^2
i3 : beta = makeGWClass M
```

```
o3 = | 1 3 |
     | 3 7 |

o3 : GrothendieckWittClass
```

Given a Grothendieck–Witt class `beta`, its underlying field can be obtained by running `getBaseField beta`, and the underlying matrix can be obtained by running either `beta.matrix` or `getMatrix beta`. Objects of type `GrothendieckWittClass` can be added and multiplied via the `addGW` and `multiplyGW` methods, respectively.

Example 6. For any unit $a \in k^\times$, there is a symmetric bilinear form of rank one

$$\begin{aligned} \langle a \rangle \colon k \times k &\to k \\ (x, y) &\mapsto axy. \end{aligned}$$

Via the change of basis $(x, y) \mapsto (bx, by)$ for any unit $b \in k^\times$, we observe that $\langle a \rangle = \langle ab^2 \rangle$. Hence the representative for a rank one form is determined only by its square class.

The following classical result (see [14, Corollary I.2.4]) implies that the forms $\langle a \rangle$ generate $\mathrm{GW}(k)$.

Theorem 7. *Every class in* $\mathrm{GW}(k)$ *is represented by a diagonal Gram matrix.*

A diagonal representative of a class in $\mathrm{GW}(k)$ can be found using the `getDiagonalClass` method.

```
i4 : getDiagonalClass beta

o4 = | 1  0 |
     | 0 -2 |

o4 : GrothendieckWittClass
```

We also provide methods for constructing various forms. We can construct a class corresponding to a list of diagonal entries via the `makeDiagonalForm` method. We denote by $\langle a_1, \ldots, a_n \rangle$ the direct sum of the rank one forms $\langle a_i \rangle$ for $1 \leq i \leq n$.

```
i5 : makeDiagonalForm(GF(13), (2,6))

o5 = | 2 0 |
     | 0 6 |

o5 : GrothendieckWittClass
```

Hyperbolic forms are crucial objects of study due to their local-to-global behavior (see Theorem 9), and they can be produced via the `makeHyperbolicForm` method. Similarly, Pfister forms, which are important objects of study in the world of quadratic forms [14, Chapter X], can be produced via the `makePfisterForm` method.

2.1. **Verifying isomorphisms of forms.** Given two non-degenerate symmetric bilinear forms, a natural question is whether they represent the same element of $\mathrm{GW}(k)$. An easy invariant to check is whether they are defined on vector spaces of the same dimension, i.e., whether the *rank* of the forms (the rank of their Gram matrices) agrees. As mentioned in Example 5, since $\mathbb{C}$ is algebraically closed and every number is a square, rank completely classifies symmetric bilinear forms over the complex numbers.

Since there are two square classes over the real numbers, namely $+1$ and -1, we can find a Gram matrix representative of any form which is diagonal, with only ± 1 appearing along the diagonal. The trace of such a Gram matrix is an invariant of the form, called the *signature*. Rank and signature jointly classify symmetric bilinear forms over the real numbers.

```
i6 : gamma = makeGWClass matrix(RR, {{3,0,0},{0,-4,0},{0,0,7}});
i7 : getSignature gamma
o7 = 1
```

Over finite fields, the *discriminant*, which is the determinant of any Gram matrix representative (valued in square classes), and the rank jointly classify symmetric bilinear forms.

Over the rational numbers, the classification of symmetric bilinear forms is more complicated. The isomorphism class of a non-degenerate form β is determined by its rank, discriminant $d_\beta \in \mathbb{Q}^*/(\mathbb{Q}^*)^2$, signature (as a form over $\mathbb{R}$), and Hasse–Witt invariants at all finite primes p. For a prime p, the Hasse–Witt invariant [17, III5, p.79] is defined as follows.

Definition 8. Given any form $\beta \cong \langle a_1, \ldots, a_n\rangle \in \mathrm{GW}(\mathbb{Q})$, its *Hasse–Witt invariant* $\varepsilon_p(\beta)$ at a prime p is the product

$$\prod_{i<j}(a_i, a_j)_p,$$

where $(-,-)_p$ denotes the *Hilbert symbol*

$$(a,b)_p := \begin{cases} 1 & \text{if } z^2 = ax^2 + by^2 \text{ has a nonzero solution in } \mathbb{Q}_p, \\ -1 & \text{otherwise.} \end{cases}$$

We can compute the Hilbert symbol $(a,b)_p$ via `getHilbertSymbol(a,b,p)` and the Hasse–Witt invariant of a form `beta` at p by `getHasseWittInvariant(beta, p)`. If p is an odd prime and the p-adic valuations $\nu_p(a)$ and $\nu_p(b)$ are even, then $(a,b)_p = 1$. Thus, $\varepsilon_p(\beta)$ is 1 for almost all primes p and only needs to be computed for $p = 2$ and odd primes p with $\nu_p(d_\beta)$ odd.

These methods together form one of our core methods `isIsomorphicForm`, which is a Boolean-valued method that determines whether two symmetric bilinear forms are isomorphic. This is done by reference to the relevant invariants over $\mathbb{C}$, $\mathbb{R}$, $\mathbb{F}_q$ (for q odd), or $\mathbb{Q}$.

2.2. **Decomposing forms.** *Witt's Decomposition Theorem* (see [14, I.4.1]) implies that any non-degenerate symmetric bilinear form decomposes into an anisotropic part and an isotropic part that is a sum of hyperbolic forms. This decomposition is crucial in simplifying an element of $\mathrm{GW}(k)$. While this decomposition is fairly routine over $\mathbb{C}$, $\mathbb{R}$, and $\mathbb{F}_q$, to decompose forms over $\mathbb{Q}$ we must implement existing algorithms from the literature. An important mathematical stepping stone is the following local-to-global principle for isotropy, a reference for which is [14, VI.3.1].

Theorem 9 (Hasse–Minkowski Principle)**.** *A form $\beta \in \mathrm{GW}(\mathbb{Q})$ is isotropic if and only if it is isotropic over $\mathbb{R}$ and over $\mathbb{Q}_p$ for all primes p.*

Our method `getAnisotropicDimensionQQp`, an implementation of [12, Algorithm 8], determines the dimension of the anisotropic part of a symmetric bilinear form over $\mathbb{Q}_p$. The method `getAnisotropicDimension` returns the anisotropic dimension of a form defined over the real numbers, the complex numbers, a finite field, or the rational numbers.

Given a form, we can therefore decompose it as

$$\beta \cong \beta_a \oplus n\mathbb{H},$$

where β_a is anisotropic, $\mathbb{H}$ denotes the hyperbolic form $\langle 1, -1\rangle$, and n is the *Witt index* (implemented as `getWittIndex`).

The Boolean-valued method `isAnisotropic` returns whether a form is anisotropic; the method `isIsotropic` is its negation.

```
i8 : alpha = makeDiagonalForm(QQ, (1,2,-3));
i9 : isAnisotropic alpha
o9 = false
i10 : isIsotropic alpha
o10 = true
```

Over $\mathbb{Q}$, the computation of the anisotropic part of β is carried out inductively by reduction of the anisotropic dimension of β, following recently published algorithms for quadratic forms over number fields by Koprowski and Rothkegel [13]. The anisotropic part of a form can be computed via `getAnisotropicPart`.

```
i11 : beta = makeDiagonalForm(QQ, (3,-3,2,5,1,-9));
i12 : getAnisotropicPart beta
```

o12 = $\begin{pmatrix} 2 & 0 \\ 0 & 5 \end{pmatrix}$

```
o12 : GrothendieckWittClass
```

A quick string reading off the decomposition of a form can be obtained by running the `getSumDecompositionString` method.

```
i13 : getSumDecompositionString beta
o13 = 2H + <2> + <5>
```

3. $\mathbb{A}^1$-Brouwer degrees

For the symbolic computations in this section, let k be an exact field[2] of characteristic not equal to 2. The methods for computing $\mathbb{A}^1$-Brouwer degrees only work for polynomials with isolated zeros [4, Theorem 1.2].

In [4], the authors show that the local and global $\mathbb{A}^1$-Brouwer degrees of an endomorphism of affine space with isolated zeros can be expressed in terms of a bilinear form associated to the Bézoutian of the endomorphism.

More explicitly, for $f_i \in k[x_1, \ldots, x_n]$, suppose $f = (f_1, \ldots, f_n)\colon \mathbb{A}^n_k \to \mathbb{A}^n_k$ has isolated zeros. Introducing new variables $(X_1, \ldots, X_n)$ and $(Y_1, \ldots, Y_n)$, we can construct the matrix Δ with entries

$$\Delta_{i,j} = \frac{f_i(Y_1, \ldots, Y_{j-1}, X_j, \ldots, X_n) - f_i(Y_1, \ldots, Y_j, X_{j+1}, \ldots, X_n)}{X_j - Y_j}.$$

One can think of this matrix Δ as a Jacobian of formal derivatives. Define $Q(f) = k[x_1, \ldots, x_n]/(f_1, \ldots, f_n)$ and $Q_p(f) = k[x_1, \ldots, x_n]_{\mathfrak{m}}/(f_1, \ldots, f_n)$ for $\mathfrak{m}$ the maximal ideal of a closed point p in the preimage of 0. The *Bézoutian* of f is defined to be the image of

[2]An *exact field* is a field whose elements are represented exactly by *Macaulay2*, e.g., $\mathbb{Q}$ or $\mathbb{F}_q$.

$\det(\Delta)$ in the algebra $Q(f) \otimes Q(f)$ (respectively, in the local algebra $Q_p(f) \otimes Q_p(f)$). Given $a_1, \ldots, a_m$ a k-linear basis of $Q(f)$ (resp., $Q_p(f)$), there are $b_{i,j} \in k$ such that

$$\det(\Delta) = \sum_{1 \leq i \leq j \leq m} b_{i,j}(a_i \otimes a_j)$$

in $Q(f) \otimes Q(f)$ (resp., $Q_p(f) \otimes Q_p(f)$). The *Bézoutian bilinear form*, the symmetric bilinear form with Gram matrix given by the $b_{i,j}$, gives the global (resp., local) $\mathbb{A}^1$-degree [4, Theorem 1.2].

In the case of the global $\mathbb{A}^1$-degree, a theorem of Macaulay tells us that a k-basis of the algebra $Q(f)$ is given by the standard monomials (see [21, Proposition 2.1]). In the case of the local $\mathbb{A}^1$-degree, a k-basis for the local ring can be calculated via the quotient of $k[x_1, \ldots, x_n]$ by the saturation of $I = (f_1, \ldots, f_n)$ at $\mathfrak{m}$.

Proposition 10 ([21, Proposition 2.5])**.** *The natural map $x_i \mapsto x_i$ defines an isomorphism of rings*

$$k[x_1, \ldots, x_n]_{\mathfrak{m}}/I \cong k[x_1, \ldots, x_n]/(I : (I : \mathfrak{m}^\infty)), \tag{2}$$

where I is an ideal of $k[x_1, \ldots, x_n]$ and $(I : (I : \mathfrak{m}^\infty))$ is the quotient of I by the saturation of I at $\mathfrak{m}$.

The `getLocalAlgebraBasis(I, m)` method uses this isomorphism to find a basis of $Q_p(f)$. It determines a k-basis of the right side of Equation (2) as a k-vector space. Proposition 10 then gives a k-basis of $Q_p(f)$.

These methods for computing k-bases for $Q(f)$ and $Q_p(f)$ allow us to algorithmically implement techniques to compute the global and local $\mathbb{A}^1$-degrees (see also [4, Section 5A]).

3.1. **A univariate polynomial.** A univariate polynomial over a field k defines an endomorphism of affine space $\mathbb{A}^1_k \to \mathbb{A}^1_k$. Consider the endomorphism $f\colon \mathbb{A}^1_{\mathbb{Q}} \to \mathbb{A}^1_{\mathbb{Q}}$ defined by

$$f(x) = (x^2 + x + 1)(x - 3)(x + 2).$$

We can compute the global degree.

```
i14 : R = QQ[x];
i15 : f = {x^4 - 6*x^2 - 7*x - 6};
i16 : alpha = getGlobalA1Degree f
```

$$\texttt{o16} = \begin{pmatrix} -7 & -6 & 0 & 1 \\ -6 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

```
o16 : GrothendieckWittClass
```

We can also compute the local degrees at the ideals $(x^2 + x + 1)$, $(x - 3)$, and $(x + 2)$, respectively.

```
i17 : I1 = ideal(x^2 + x + 1);
o17 : Ideal of R
i18 : alpha1 = getLocalA1Degree(f, I1)
```

$$\texttt{o18} = \begin{pmatrix} -5 & -7 \\ -7 & -2 \end{pmatrix}$$

```
o18 : GrothendieckWittClass
i19 : I2 = ideal(x - 3)
o19 = ideal (x - 3)
o19 : Ideal of R
i20 : alpha2 = getLocalA1Degree(f, I2)
o20 = ( 65 )
o20 : GrothendieckWittClass
i21 : I3 = ideal(x + 2);
o21 : Ideal of R
i22 : alpha3 = getLocalA1Degree(f, I3)
o22 = ( -15 )
o22 : GrothendieckWittClass
```

We can then use the isIsomorphicForm method (see also Section 2.1) to verify that the local $\mathbb{A}^1$-degrees sum to the global $\mathbb{A}^1$-degree.

```
i23 : alpha' = addGW(alpha1,addGW(alpha2,alpha3));
i24 : isIsomorphicForm(alpha,alpha')
o24 = true
```

Consider the graph of $f(x)$.



Following [16, Theorem 1.2], $\mathbb{A}^1$-degrees can be understood as enriched intersection numbers, determined by the signed volume of the parallelepiped spanned by the gradient vectors of the hypersurfaces at the intersection point. In the one-dimensional case, considering the normal vectors, we can interpret $\alpha_2 = \langle 65 \rangle$, the local $\mathbb{A}^1$-degree at $(x-3)$, and $\alpha_3 = \langle -15 \rangle$, the local $\mathbb{A}^1$-degree at $(x+2)$, as signs of the derivative at these points.

3.2. **The Euler characteristic of the Grassmannian of lines in $\mathbb{P}^3$.** For k a field of characteristic not 2, let $\mathrm{Gr}_k(2,4)$ be the Grassmannian of lines in $\mathbb{P}^3_k$. Following [4, Example 8.2], we can compute the $\mathbb{A}^1$-Euler characteristic of the Grassmannian over $k = \mathbb{F}_{27}$ as the $\mathbb{A}^1$-degree of the section $\sigma\colon \mathbb{A}^4_{\mathbb{F}_{27}} \to \mathbb{A}^4_{\mathbb{F}_{27}}$ defined by[3]

$$(x_1, x_2, x_3, x_4) \mapsto (x_2 - x_1x_3, 1 - x_1x_4, x_4 - x_1 - x_3^2, -x_2 - x_3x_4).$$

[3]There is a small error in the definition of σ in [4, Example 8.2]. The second and third component functions of σ should be swapped in order to agree with the ordered basis induced on the tangent bundle of $\mathrm{Gr}_k(2,4)$ as in [11, Proposition 45]. By [4, Example 6.3], the overall computation is only affected by a sign.

We compute the $\mathbb{A}^1$-Euler characteristic as follows.

```
i25 : k = GF(27);
i26 : x = symbol x;
i27 : R = k[x_1,x_2,x_3,x_4];
i28 : f = {x_2 - x_1*x_3, 1 - x_1*x_4, x_4 - x_1 - x_3^2, -x_2 - x_3*x_4};
i29 : beta = getGlobalA1Degree f
```

$$
o29 = \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}
$$

```
o29 : GrothendieckWittClass
```

We can subsequently use the `getSumDecompositionString` method to decompose the symmetric bilinear form β.

```
i30 : getSumDecompositionString beta
o30 = 2H + <1> + <1>
```

Our computation agrees with the result given in [4, Example 8.2] and shows

$$\chi(\mathrm{Gr}_{\mathbb{F}_{27}}(2,4)) = 2\mathbb{H} + \langle 1\rangle + \langle 1\rangle.$$

3.3. **Local geometry of some lines on the Fermat cubic surface.** In their pioneering paper [11], Kass and Wickelgren give a Grothendieck–Witt class-valued count of the number of lines on a smooth cubic surface, providing an interpretation of the local $\mathbb{A}^1$-degree as the topological type of the line. To illustrate some features of the `A1BrouwerDegrees` package, we use it to compute the topological type of some lines on the Fermat cubic surface.

Let k be a field, and let $\{e_1, e_2, e_3, e_4\}$ be the standard basis for k^4. By [11, Lemma 45], we can define local coordinates on $\mathrm{Spec}(k[y_1, y_2, y_3, y_4]) \cong \mathbb{A}^4_k$ around the point of $\mathrm{Gr}_k(2,4)$ defined by the span of $\{e_3, e_4\}$ such that y_1, y_2, y_3, y_4 corresponds to the span of $\{\widetilde{e}_3, \widetilde{e}_4\}$, where

$$\widetilde{e}_i = \begin{cases} e_i & \text{for } i \in \{1,2\}, \\ e_1y_1 + e_2y_2 + e_3 & \text{for } i = 3, \\ e_1y_3 + e_2y_4 + e_4 & \text{for } i = 4. \end{cases}$$

Letting $\mathcal{S}$ denote the tautological bundle over $\mathrm{Gr}_k(2,4)$, the above coordinates provide a trivialization of the vector bundle $\mathrm{Sym}^3 \mathcal{S}^\vee$ over the open affine subvariety

$$U \cong \mathrm{Spec}(k[y_1, y_2, y_3, y_4]) \subseteq \mathrm{Gr}_k(2,4).$$

A cubic surface X defines a section $\sigma_X|_U : U \to \mathrm{Sym}^3 \mathcal{S}^\vee|_U$ that vanishes on the lines on X that, when treated as affine two-dimensional subspaces of k^4, contain e_3 and e_4 in their span.

Let us consider the Fermat cubic surface defined by the homogeneous cubic equation $x_0^3 + x_1^3 + x_2^3 + x_3^3$. That is,

$$X = \left\{[x_0 : x_1 : x_2 : x_3] \in \mathbb{P}^3_k : x_0^3 + x_1^3 + x_2^3 + x_3^3 = 0\right\} \subseteq \mathbb{P}^3_k.$$

Working over $\mathbb{Q}$, the lines on X are all defined over the cyclotomic extension $\mathbb{Q}(\zeta)$ for ζ a primitive third root of unity. We can explicitly compute the 27 lines as

$$[s : t : -\zeta^i t : -\zeta^j s], [s : t : -\zeta^i s : -\zeta^j t], [s : -\zeta^i s : t : -\zeta^j t]$$

for $0 \leq i, j \leq 2$ and $[s : t] \in \mathbb{P}^1_{\mathbb{Q}}$. Note that there are only 18 lines containing e_3 and e_4 in their span. We thus expect the section to vanish at 18 points. Applying Pauli's computation of sections of $\sigma_X|_U : U \to \mathrm{Sym}^3 \mathcal{S}^\vee|_U$ in [19, §2.2, Remark 2.7], our section is of the form $\sigma_X|_U = (f_1, f_2, f_3, f_4)$ where

$$\begin{aligned} f_1(y_1, y_2, y_3, y_4) &= y_1^3 + y_3^3 + 1 \\ f_2(y_1, y_2, y_3, y_4) &= 3y_1^2 y_2 + 3y_3^2 y_4 \\ f_3(y_1, y_2, y_3, y_4) &= 3y_1 y_2^2 + 3y_3 y_4^2 \\ f_4(y_1, y_2, y_3, y_4) &= y_2^3 + y_4^3 + 1. \end{aligned}$$

We compute the global $\mathbb{A}^1$-degree, which is rank 18, as expected.

```
i31 : R = QQ[y_1,y_2,y_3,y_4];
i32 : f = {y_1^3 + y_3^3 + 1,
          3*y_1^2*y_2 + 3*y_3^2*y_4,
          3*y_1*y_2^2 + 3*y_3*y_4^2,
          y_2^3 + y_4^3 + 1};
i33 : alpha = getGlobalA1Degree f;
i34 : getSumDecompositionString alpha
o34 = 8H + <1> + <1>
```

To compute the local degree, we find an isolated zero using the `minimalPrimes` method of *Macaulay2*.

```
i35 : I = (minimalPrimes ideal f)_0
o35 = ideal (y_4, y_3 + 1, y_2 + 1, y_1)
o35 : Ideal of R
```

We then compute the local $\mathbb{A}^1$-degree at this point.

```
i36 : beta = getLocalA1Degree(f, I)
o36 = ( 81 )
o36 : GrothendieckWittClass
i37 : getSumDecomposition beta
o37 = ( 1 )
o37 : GrothendieckWittClass
```

At the other nine minimal primes, the same calculation gives the local degrees as one copy of $\langle 1 \rangle$, six copies of $\langle 3, -1 \rangle$ and two copies of $\langle 2, -6 \rangle$. As

$$6\langle 3, -1 \rangle + 2\langle 2, -6 \rangle \cong 8\mathbb{H},$$

the global $\mathbb{A}^1$-degree equals the sum of the local $\mathbb{A}^1$-degrees.

The local computation above indicates that the line spanned by $\{-e_2 + e_3, -e_1 + e_4\}$ on the Fermat cubic surface is a hyperbolic line. We briefly show this agrees with the type as defined in [11, Definition 9].

By [11, Proposition 14], the local type of the line is equal to the resultant of the partial derivatives of the equation of the Fermat cubic surface restricted to the line. Letting z_1, z_2, z_3, z_4 be the dual basis to e_1, e_2, e_3, e_4 defined above, we can write the equation of the Fermat surface in terms of the dual basis via the change of basis $z_1 \mapsto z_1 + z_4, z_2 \mapsto z_2 + z_3, z_3 \mapsto -z_3, z_4 \mapsto -z_4$ so that the line is spanned by e_3 and e_4.

```
i38 : needsPackage "Resultants";
i39 : R = QQ[z_1,z_2][z_3,z_4];
i40 : fermat = (z_1 + z_4)^3 + (z_2 + z_3)^3 - z_3^3 - z_4^3;
```

We compute the restriction of the partial derivatives of the defining equation of the Fermat cubic surface to the surface with respect to the dual basis z_1, z_2.

```
i41 : g1 = sub(diff(z_1, fermat), {z_1 => 0, z_2 => 0});
i42 : g2 = sub(diff(z_2, fermat), {z_1 => 0, z_2 => 0});
```

We then compute the resultant of these polynomials and consider it as a quadratic form over $\mathbb{Q}$ in order to agree with the computation of the local index over $\mathbb{Q}$.

```
i43 : line_type = makeDiagonalForm(QQ, lift(resultant {g1,g2}, QQ))
o43 = ( 81 )
o43 : GrothendieckWittClass
i44 : isIsomorphicForm(line_type,beta)
o44 = true
```

Thus this computation agrees with the local $\mathbb{A}^1$-degree of the associated section of $\mathrm{Sym}^3 \mathcal{S}^*$ as computed above.

Acknowledgements

We would like to thank the organizers of the 2023 *Macaulay2* Workshop for the opportunity to work together developing this package. The second-named author would like to thank Sabrina Pauli for helpful conversations about this work. We would also like to thank the anonymous referees for detailed comments that helped to improve both this paper and the code. The second-named author is supported by an NSF Postdoctoral Research Fellowship (DMS-2303242), and the sixth-named author is partially supported by an NSF Standard Grant of Frank Sottile (DMS-2201005).

References

[1] D. Agostini and M. Kummer. Secant varieties of curves, Ulrich bundles and the arithmetic writhe, 2023. arXiv:2307.07543 [math.AG].

[2] T. Bachmann and K. Wickelgren. Euler classes: six-functors formalism, dualities, integrality and linear subspaces of complete intersections. *J. Inst. Math. Jussieu*, 22(2):681–746, 2023.

[3] T. Brazelton. An introduction to $\mathbb{A}^1$-enumerative geometry. In *Homotopy theory and arithmetic geometry—motivic and Diophantine aspects*, volume 2292 of *Lecture Notes in Math.*, pages 11–47. Springer, Cham, 2021.

[4] T. Brazelton, S. McKean, and S. Pauli. Bézoutians and the $\mathbb{A}^1$-degree. *Algebra Number Theory*, 17(11):1985–2012, 2023.

[5] D. Eisenbud and H. I. Levine. An algebraic formula for the degree of a C^∞ map germ. *Ann. of Math. (2)*, 106(1):19–44, 1977. With an appendix by Bernard Teissier, "Sur une inégalité à la Minkowski pour les multiplicités".

[6] D. R. Grayson and M. E. Stillman. Macaulay2, a software system for research in algebraic geometry. Available at http://www2.macaulay2.com.

[7] G. N. Himšiašvili. The local degree of a smooth mapping. *Sakharth. SSR Mecn. Akad. Moambe*, 85(2):309–312, 1977.

[8] A. Jaramillo Puentes and S. Pauli. A quadratically enriched correspondence theorem, 2024. arXiv:2309.11706 [math.AG].

[9] J. L. Kass, M. Levine, J. P. Solomon, and K. Wickelgren. A quadratically enriched count of rational curves, 2023. arXiv:2307.01936 [math.AG].

[10] J. L. Kass and K. Wickelgren. The class of Eisenbud-Khimshiashvili-Levine is the local $\mathbf{A}^1$-Brouwer degree. *Duke Math. J.*, 168(3):429–469, 2019.

[11] J. L. Kass and K. Wickelgren. An arithmetic count of the lines on a smooth cubic surface. *Compos. Math.*, 157(4):677–709, 2021.

[12] P. Koprowski and A. Czogała. Computing with quadratic forms over number fields. *J. Symbolic Comput.*, 89:129–145, 2018.

[13] P. Koprowski and B. Rothkegel. The anisotropic part of a quadratic form over a number field. *J. Symbolic Comput.*, 115:39–52, 2023.

[14] T. Y. Lam. *Introduction to quadratic forms over fields*, volume 67 of *Graduate Studies in Mathematics*. American Mathematical Society, Providence, RI, 2005.

[15] M. Levine. Aspects of enumerative geometry with quadratic forms. *Doc. Math.*, 25:2179–2239, 2020.

[16] S. McKean. An arithmetic enrichment of Bézout's Theorem. *Math. Ann.*, 379(1–2):633–660, 2021.

[17] J. Milnor and D. Husemoller. *Symmetric bilinear forms*, volume Band 73 of *Ergebnisse der Mathematik und ihrer Grenzgebiete [Results in Mathematics and Related Areas]*. Springer-Verlag, New York-Heidelberg, 1973.

[18] F. Morel. *$\mathbb{A}^1$-algebraic topology over a field*, volume 2052 of *Lecture Notes in Mathematics*. Springer, Heidelberg, 2012.

[19] S. Pauli. Computing $\mathbb{A}^1$-Euler numbers with Macaulay2. *Res. Math. Sci.*, 10(3):Paper No. 26, 14 pp., 2023.

[20] S. Pauli and K. Wickelgren. Applications to $\mathbb{A}^1$-enumerative geometry of the $\mathbb{A}^1$-degree. *Res. Math. Sci.*, 8(2):Paper No. 24, 29 pp., 2021.

[21] B. Sturmfels. *Solving systems of polynomial equations*, volume 97 of *CBMS Regional Conference Series in Mathematics*. Published for the Conference Board of the Mathematical Sciences, Washington, DC; by the American Mathematical Society, Providence, RI, 2002.

(N. Borisov) Department of Mathematics, University of Pennsylvania, Philadelphia, PA 19104, USA
Email address: nborisov@sas.upenn.edu
URL: https://nikita-borisov.github.io/

(T. Brazelton) Department of Mathematics, Harvard University, Cambridge, MA 02138, USA
Email address: brazelton@math.harvard.edu
URL: https://tbrazel.github.io/

(F. Espino) Department of Mathematics, University of Pennsylvania, Philadelphia, PA 19104, USA
Email address: frenly@sas.upenn.edu
URL: https://frenlye.github.io/

(T. Hagedorn) Department of Mathematics and Statistics, The College of New Jersey, Ewing, NJ 08618, USA
Email address: hagedorn@tcnj.edu
URL: https://hagedorn.pages.tcnj.edu/

(Z. Han) University of Pennsylvania, Philadelphia, PA 19104, USA
Email address: zbtomhan@sas.upenn.edu
URL: https://www.linkedin.com/in/zhaobo-han-77b1301a2/

(J. Lopez Garcia) Department of Mathematics, Texas A&M University, College Station, TX 77843, USA
Email address: jordy.lopez@tamu.edu
URL: https://jordylopez27.github.io/

(J. Louwsma) Department of Mathematics, Niagara University, Niagara University, NY 14109, USA
Email address: jlouwsma@niagara.edu
URL: https://www.joellouwsma.com

(W. J. G. Ong) Bowdoin College, Brunswick, ME 04011, USA
Email address: gong@bowdoin.edu
URL: https://wgabrielong.github.io/

(A. R. Tawfeek) Department of Mathematics, University of Washington, Seattle, WA 98195, USA
Email address: atawfeek@uw.edu
URL: https://atawfeek.com/